

# Timed Strategies for Real-Time Rewrite Theories

Carlos Olarte<sup>1</sup> and Peter Csaba Ölveczky<sup>2</sup>

<sup>1</sup> LIPN, CNRS UMR 7030, Université Sorbonne Paris Nord, France

<sup>2</sup> Department of Informatics, University of Oslo, Oslo, Norway

**Abstract.** In this paper we propose a language for conveniently defining a wide range of execution strategies for real-time rewrite theories, and provide Maude-strategy-implemented versions of most Real-Time Maude analysis methods, albeit with user-defined discrete and timed strategies. We also identify a new time sampling strategy that should provide both efficient and exhaustive analysis for many distributed real-time systems. We exemplify the use of our language and its analyses on a simple round trip time protocol, and compare the performance of standard Maude search with our strategy-implemented reachability analyses on the CASH scheduling algorithm benchmark.

## 1 Introduction

Real-time systems can naturally be defined in rewriting logic [22] as *real-time rewrite theories* [29]. In such theories, actions that can be assumed to take zero time are modeled by ordinary (also called *instantaneous*) rewrite rules, and time advance is modeled by labeled “tick” rewrite rules of the form  $[l] : \{t_1\} \rightarrow \{t_2\}$  **in time**  $\tau$  **if**  $cond$ , where the *whole* system state has the form  $\{t\}$ .

Real-time rewrite theories inherit the expressiveness and modeling convenience of rewriting logic, and allow us to model a wide range of distributed real-time systems—with different communication forms, user-defined data types, dynamic object creation and deletion, and so on—in an object-oriented style.

The specification and analysis of real-time rewrite theories is supported by the Real-Time Maude [31,32,26] language and tool, which is implemented in Maude as an extension of Full Maude [12]. For dense time, the tick rules typically have the form `cr1 [tick] : {t} => {u} in time T if T <= f(t)`, where  $T$  is a variable of sort `Time` not appearing in the term  $t$  [31].

Real-Time Maude provides *explicit-state* analysis methods, where the above tick rules are executed according to a *time sampling strategy*, where the variable  $T$  in the rule above is instantiated either to:

- a user-selected value (such as 1); or to
- the maximal possible time increase  $f(t)$  (or a value  $\Delta$  if  $f(t)$  is infinity).

Real-Time Maude supports unbounded and time-bounded reachability analysis, LTL and timed CTL model checking, and other time-specific analyses. All such explicit-state analyses are performed with the selected time sampling strategy.

The cost of the expressiveness and generality of Real-Time Maude is that time-sampling-based analysis is not sound in general for dense time [30]. Nevertheless, counterexamples/reachable states obtained by Real-Time Maude analysis are real counterexamples/reachable states. Real-Time Maude has therefore been used to discover subtle but significant bugs in a number of sophisticated state-of-the-art systems beyond the scope of decidable formalisms like timed automata, including: a 50-page active network protocol [33] (which required advanced functions and detailed modeling of communication), state-of-the-art wireless sensor network algorithms [34] (which required modeling coverage areas, angles, etc., and functions on such), mobile ad-hoc network leader election protocols [20] (the fault was due to a subtle interplay between node movements and communication delays), scheduling algorithms with reuse of unused budgets [28] (which somewhat surprisingly required unbounded queues), a traffic intersection system from the Ptolemy II library [5,18] (which required defining the semantics of Ptolemy II discrete-event models), cloud-based transaction systems [16,8], and an error in running cars which was not found by other methods, and where reportedly Real-Time Maude time sampling was key.<sup>3</sup>

Real-Time Maude’s expressiveness and generality also makes it a suitable semantic framework and formal analysis backend in which (subsets of) modeling languages such as AADL [27], Ptolemy II DE models [5], Timed Rebeca [39], and a DoCoMo Labs handset language [1] have been given a formal semantics and formal analysis capabilities [25].

Maude was recently equipped with a strategy language to allow users to define specific execution strategies on top of their Maude specifications [15]. In this paper we use Maude’s strategy language to define useful strategies for real-time rewrite theories. This work is motivated by the following issues:

1. Real-Time Maude analyses apply one of the above time sampling strategies to *all* applications of tick rules. However, as the following example shows, more sophisticated time sampling strategies are often desired:

Consider a system computing the *round trip time* (RTT) between two nodes every five seconds. A time sampling strategy that visits every time unit covers all possible behaviors in discrete time domains, but visits each time point even when the round trip time in that round already has been found, leading to *very inefficient* analyses. On the other hand, increasing time maximally in each tick step only takes into account those behaviors where each message has been delayed as much as possible, thereby always finding the same (maximal) round trip time value, which is not very helpful.

In this simple but prototypical example, the ideal time sampling strategy advances time by one time unit as long as there is a message (which could arrive at “any” time) in the state, and increases time maximally when there is no message in the state (and we are just idling until the next iteration of the RTT protocol). Such time sampling would cover all possible behaviors, yet would not stop time unnecessarily.

---

<sup>3</sup> Hitoshi Ohsaki, personal communication, 2007.

2. The user may also want to define her own *non-time-sampling* (called *discrete*) execution strategies (e.g., what happens if we restrict the model of a coffee machine so that the user can add at most two spoons of sugar?). We should therefore support the analysis of real-time systems with user-defined execution strategies on the non-time parts, combined with the desired (possibly user-defined) time sampling strategy. This includes typical “timed strategies” in real-time systems including: *eagerness* of all/some actions; given priority to some actions; deciding the next action according to the previously states visited; and so on. Such more advanced strategies cannot be easily defined on top of the current implementation of Real-Time Maude.
3. Maude’s strategy language has an efficient implementation, using multi-threading and the option of depth-first search analyses, which provides better performance than standard Maude search in some cases, as shown in Section 5.6.
4. Real-Time Maude is implemented as an extension of Full Maude. While this allowed us to use Full Maude’s infrastructure to define distributed real-time systems in a very useful object-oriented style, extending the large Full Maude implementation posed challenges in terms of integrating other analysis methods for real-time rewrite theories into Real-Time Maude, and of maintaining and upgrading the tool to newer versions of Full Maude. Since Maude 3.3 directly supports object-oriented specification, and since Full Maude will no longer be maintained and upgraded, we are currently working on developing the next version of Real-Time Maude as a much “lighter” Maude implementation that does not extend Full Maude. In this context, doing as much as possible as easily as possible using available Maude features is needed.

In this paper we therefore show how most Real-Time Maude analysis methods can be performed by rewriting with strategies directly in Maude (Section 4).

However, even those analysis methods needed somewhat hard-to-understand strategy expressions, even without advanced time sampling strategies. This begs the question how the casual Maude user can analyze her system with more complex discrete and time sampling strategies. Furthermore, since (Real-Time) Maude can be used as a formal analysis backend for many real-time modeling languages, we really need an intuitive way to define useful strategies for the non-expert Maude user. For example, in [3,4] we claim as one of the main selling point of our Maude framework, compared to state-of-the-art domain-specific tools for parametric timed automata and time Petri nets, the ability to analyze the system with user-defined execution strategies. However, this selling point becomes moot if the timed automata/Petri net expert cannot define her strategies.

To address this issue, in Section 5 we define and implement what we hope is an intuitive and fairly powerful timed strategy language for real-time rewrite theories. This language should make it easy for the casual user to define a wide range of useful discrete strategies as well as advanced state- and even history-dependent time sampling strategies in a modular way.

Even the most intuitive language would be useless if systems cannot be analyzed efficiently. In Section 5.6 we therefore compare the performance of standard

Maude search with our strategy-implemented analysis methods on a sophisticated scheduling system [28]. It turns out that using a depth-search strategy on this system sometimes leads to a faster analysis.

Finally, we discuss related work in Section 6, and give some concluding remarks in Section 7. The strategy language and its implementation, with Maude models and their strategies and execution commands are available at [24].

## 2 Preliminaries

*Rewriting Logic and Maude.* Maude [13] is a rewriting-logic-based executable formal specification language and high-performance analysis tool for distributed systems. A Maude module specifies a *rewrite theory*  $(\Sigma, E \cup A, R)$ , where:

- $\Sigma$  is an algebraic *signature*; i.e., a set of *sorts*, *subsorts*, and *function symbols*.
- $(\Sigma, E \cup A)$  is a *membership equational logic* [23] theory, with  $E$  a set of possibly conditional equations and membership axioms, and  $A$  a set of equational axioms such as associativity, commutativity, and identity, so that equational deduction is performed *modulo* the axioms  $A$ .
- $R$  is a collection of *labeled conditional rewrite rules*  $[l] : t \longrightarrow t' \text{ if } cond$ , specifying the system's local transitions.

A function  $f$  is declared `op  $f : s_1 \dots s_n \rightarrow s$` . Equations and rewrite rules are introduced with, respectively, keywords `eq`, or `ceq` for conditional equations, and `rl` and `crl`. Mathematical variables are declared with the keywords `var` and `vars`, or can have the form `var : sort` and be introduced on the fly.

A declaration `class  $C \mid att_1 : s_1, \dots, att_n : s_n$`  declares a *class*  $C$  of objects with attributes  $att_1$  to  $att_n$  of sorts  $s_1$  to  $s_n$ . An *object instance* of class  $C$  is represented as a term  $\langle O : C \mid att_1 : val_1, \dots, att_n : val_n \rangle$ , where  $O$ , of sort `Obj`, is the object's *identifier*, and where  $val_1$  to  $val_n$  are the current values of the attributes  $att_1$  to  $att_n$ . A *message* is a term of sort `Msg`. A system state is modeled as a term of the sort `Configuration`, and has the structure of a *multiset* made up of objects and messages.

The dynamic behavior of a system is axiomatized by specifying each of its transition patterns by a rewrite rule. For example, the rule (with label 1)

$$\begin{aligned} \text{rl } [1] : & \langle 0 : C \mid a1 : f(x, y), a2 : 0', a3 : z \rangle \\ \Rightarrow & \langle 0 : C \mid a1 : x + z, a2 : 0', a3 : z \rangle . \end{aligned}$$

defines a family of transitions in which the attribute `a1` of object `0` is updated to `x+z`. Attributes whose values do not change and do not affect the next state, such as `a2` and the right-hand side occurrence of `a3`, need not be mentioned.

*Formal Analysis in Maude.* Maude provides a number of analysis methods, including rewriting for simulation purposes, reachability analysis, and linear temporal logic (LTL) model checking. The command `red expr` reduces the expression *expr* to its normal form using the equations  $E$ . The rewrite command `frew init` simulates one behavior from the initial state/term *init* by applying rewrite rules.

Given a state pattern *pattern* and an (optional) condition *cond*, Maude’s `search` command searches the reachable state space from *init* for all (or optionally a given number of) states that match *pattern* such that *cond* holds:

```
search init =>* pattern [such that cond] .
```

*Strategies.* Maude provides a language for defining strategies to control and restrict rewriting. A strategy may not make rewriting deterministic, and hence multiple behaviors allowed by the strategy must be explored. The Maude command `srew t using str` rewrites the term *t* according to the strategy *str*, and returns a set of terms, possibly bounded by the number of desired solutions. `srew` explores multiple paths in parallel, and ensures that solutions will eventually be found. The command `dsrew t using str` explores the behaviors allowed by *str* in a depth-first way.

Basic rewrite strategies *str* include `l[σ]` (apply rule labeled *l* once with the optional substitution  $\sigma$ ), `all` (apply any of the rules, except those marked `nonexec`, once), `idle` (identity), `fail` (empty set), and `match P s.t. C`, which checks whether the current term matches the pattern *P* subject to the constraint *C*. Compound strategies can be defined using concatenation ( $\alpha ; \beta$ ), disjunction ( $\alpha | \beta$ , whose result is the union of the results of  $\alpha$  and  $\beta$ ), iteration ( $\alpha^*$ ),  `$\alpha$  or-else  $\beta$`  (execute  $\alpha$ , and  $\beta$  if  $\alpha$  fails), `try( $\alpha$ )` (applies  $\alpha$  if it does not fail), normalization  `$\alpha!$`  (execute  $\alpha$  until it cannot be further applied), `matchrew  $p(x_1, \dots, x_n)$  s.t. cond by  $x_1$  using  $\alpha_1, \dots, x_n$  using  $\alpha_n$`  (if the term matches the pattern  $p(x_1, \dots, x_n)$ , then, for each match  $\sigma$ , rewrite each substitution instance  $x_i\sigma$  in the term according to the strategy  $\alpha_i$ ), and so on [12, chapter 10].

*Metaprogramming.* Maude supports metaprogramming in the sense that a Maude specification *M* can be represented as a term  $\overline{M}$  (of sort `Module`), and a term *t* in *M* can be (meta-)represented as a term  $\overline{t}$  of sort `Term`. Obtaining the meta-representations of a module *M* and a term *t* in Maude is possible using the built-in functions `upModule(M)` and `upTerm(t)` respectively. Maude’s `META-LEVEL` module contains a number of useful meta-level versions of key Maude functionality, including `metaSrewrite` (`srew` and `dsrew` at the meta-level).

*Real-Time Rewrite Theories and Real-Time Maude.* Real-time systems can be defined in rewriting logic as real-time rewrite theories [29], which are parametric in the (discrete or dense) time domain. In such theories, ordinary rewrite rules model *instantaneous* change, and time advance is modeled explicitly by “tick” rewrite rules of the form `cr1 [tick] {t} => {u} in time  $\tau$  if cond`, where  $\tau$  is a term of sort `Time`, *t* and *u* are terms of sort `System`, and where the *entire* state always has the form `{s}`, so that `{_}` does not occur in *s*; this ensures that time advances uniformly in the whole system.

Real-Time Maude [31,32,26] supports the modeling and analysis of real-time rewrite theories. Most Real-Time Maude specifications (not only for dense time) have tick rules of the form

```
cr1 [tick] : [t( $\bar{x}$ )] => {u( $\bar{x}, y$ )} in time y if y <= f(t( $\bar{x}$ )) /\ cond [nonexec] .
```

where  $y$  is a variable that does not appear in  $t$  and is not instantiated in  $cond$ , making the rule non-executable (**nonexec**) as it stands.

Real-Time Maude therefore offers the user the possibility of choosing between the following *time sampling strategies* for executing such *time-nondeterministic* tick rewrite rules:

- *deterministic time sampling* with a user-given time value  $\delta > 0$ ; and
- *maximal time sampling*, with a user-given “default” time value  $\delta > 0$ .

Using *deterministic* time sampling, the variable  $y$  in the above tick rule is instantiated by the selected time value  $\delta$  in each application of a time-nondeterministic tick rule; the tick rule cannot be applied to a state  $\{s\}$  if  $f(s)$  is smaller than  $\delta$ .

Using *maximal* time sampling, the variable  $y$  is instantiated to advance time as much as possible, namely, by  $f(s)$  in state  $\{s\}$ , unless  $f(s)$  is the infinity value **INF**, in which case  $y$  is instantiated with the “default” time value  $\delta$  instead. The tick rule is not applied to state  $\{s\}$  if the maximal time increase  $f(s)$  equals 0.

Real-Time Maude provides the following analysis methods, where the *same* selected time sampling strategy is applied in *all* tick rule applications [31]:

- *Rewriting up to time  $\Lambda$* .
- *Timed and untimed search* for states matching a pattern  $p(\bar{x})$ , such that an optional condition  $cond(\bar{x})$  holds, that are reachable from the initial state *init* within a given time interval  $[l, u]$  (where  $u$  could be **INF**) for time-bounded search, or in any time for untimed search.
- *Time-bounded and unbounded LTL model checking* check whether each behavior from *init*, up to a given time bound in the time-bounded case, satisfies an (untimed) linear temporal logic (LTL) formula.
- *Timed CTL model checking* checks whether each behavior, possibly up to a user-given time bound, satisfies a given *timed* CTL formula [18].
- *Find latest* finds the longest time it takes to reach a desired state.
- *Find earliest* finds the shortest time needed to find the desired state.

Since the Real-Time Maude analyses only cover the behaviors possible with the selected time sampling strategy, they may not give the correct results [28].

In time-bounded analysis in Real-Time Maude, internally the state also contains the “system clock” denoting the time it takes to reach the corresponding state; this adds significantly to the state space. In contrast, in *unbounded* analyses we do not carry this clock component with the states.

### 3 The Targeted Real-Time Rewrite Theories

This section presents some assumptions about the real-time rewrite theories we consider in the rest of this paper. These assumptions are needed to define *generic* timed strategies and strategy languages. Most, if not all, large Real-Time Maude applications naturally belong to this class of real-time rewrite theories, or can easily be modified to do so (e.g., by renaming rule labels and variables).

Section 3.2 presents our running example as one such model of a prototypical real-time system: a simple protocol for computing the *round trip times* between pairs of senders and receivers in a network.

### 3.1 Assumptions

Since we are in the process of extending and reimplementing Real-Time Maude, and want to use Maude features as much and directly as possible, we specify our real-time rewrite theories directly in Maude by extending the following “timed prelude,” which defines the sorts of our states:

```
fmod TIMED-PRELUDE is including TIME .
  sorts System GlobalSystem ClockedSystem .
  subsort GlobalSystem < ClockedSystem .

  op {} : System -> GlobalSystem [ctor] .
  op _in time_ : GlobalSystem Time -> ClockedSystem [ctor] .

  var CLS : ClockedSystem . vars T T' : Time .
  eq (CLS in time T) in time T' = CLS in time (T plus T') .
endfm
```

We assume a sort `Time` for the time values, a supersort `TimeInf` adding an infinity element `INF` to those values, and assume that each tick rule has the form

```
var T : Time .
crl [tick] : {t} => {u} in time T if T <= mte(t) /\ cond .
```

or the form

```
var T : Time .
rl [tick] : {t} => {u} in time T .
```

where the symbols in italics are placeholders for terms and conditions. In particular, we assume that the unknown time advance is represented by the specific variable `T` (not appearing in `t` nor in `cond`), that all tick rules are labeled `tick`, that no non-tick rule is labeled `tick`, and that the maximal time elapse is given by the (user-defined) function `mte`, which returns a time value or `INF`. (It is easy to define meta-level functions renaming labels and variables to conform to these requirements.)

### 3.2 Running Example: Finding Round Trip Times

The following “Real-Time Maude-style” object-oriented Maude model specifies a very simple protocol for computing the *round trip time* (RTT) between two nodes (a `sender` and a `receiver`) every 5 seconds. The delay of a message can be any value between a lower and an upper bound. This small but prototypical example contains many features of larger real-time distributed protocols: clocks, timers, and messages with nondeterministic delays.

```
load rtm-prelude
omod RTT is
```

```

including TIMED-PRELUDE .
protecting NAT-TIME-DOMAIN-WITH-INF .

var M : Msg .      var TI : TimeInf .      vars T T2 T3 : Time .
vars R S : Oid .   vars C1 C2 STATE : Configuration .

sort DlyMsg .
subsorts Msg < DlyMsg < Configuration < System .

op dly : Msg Time Time -> DlyMsg [ctor] . --- upper and lower bounds

rl [deliver] : dly(M, 0, TI) => M . --- can deliver ripe message any time

msgs rttReq_from_to_ rttResp_from_to_ : Time Oid Oid -> Msg .

class Sender | clock : Time, timer : Time, lowerDly : Time, period : Time,
              upperDly : TimeInf, rtt : TimeInf, receiver : Oid .

class Receiver | lowerDly : Time, upperDly : TimeInf .

rl [send] :
  < S : Sender | clock : T, timer : 0, period : T2,
                lowerDly : T3, upperDly : TI, receiver : R >
=>
  < S : Sender | timer : T2 >
  dly(rttReq T from S to R, T3, TI) .

rl [respond] :
  (rttReq T from S to R)
  < R : Receiver | lowerDly : T3, upperDly : TI >
=>
  < R : Receiver | >
  dly(rttResp T from R to S, T3, TI) .

rl [recordRTT] :
  (rttResp T from R to S)
  < S : Sender | clock : T2 >
=>
  < S : Sender | rtt : T2 monus T > .

crl [tick] :
  {STATE} => {timeEffect(STATE, T)} in time T
  if T <= mte(STATE) [nonexec] .

op mte : Configuration -> TimeInf [frozen] .
eq mte(none) = 0 .
ceq mte(C1 C2) = min(mte(C1), mte(C2)) if C1 /= none and C2 /= none .

eq mte(< S : Sender | timer : TI >) = TI .
eq mte(< R : Receiver | >) = INF .
eq mte(dly(M, T, TI)) = TI .
eq mte(M) = 0 . --- ripe message must be read immediately

op timeEffect : Configuration Time -> Configuration .
eq timeEffect(none, T) = none .
ceq timeEffect(C1 C2, T) = timeEffect(C1, T) timeEffect(C2, T)
  if C1 /= none and C2 /= none .
eq timeEffect(< S : Sender | clock : T, timer : TI >, T2)

```



```

= < S : Sender | clock : T + T2, timer : TI minus T2 > .
eq timeEffect(< R : Receiver | >, T) = < R : Receiver | > .
eq timeEffect(dly(M, T, TI), T2) = dly(M, T minus T2, TI minus T2) .
eq timeEffect(M, T) = M .

ops snd rcv : -> Oid [ctor] .

op init : -> ClockedSystem .
eq init
= {< snd : Sender | clock : 0, timer : 0, period : 5000, lowerDly : 5,
    upperDly : 20, rtt : INF, receiver : rcv >
  < rcv : Receiver | lowerDly : 7, upperDly : 30 >} in time 0 .
endop

```

Each **Sender** object has the following attributes: **clock** denotes its “local clock” (used to compute the round trip time); **timer** denotes the time until the next round begins; **lowerDelay** and **upperDelay** give the bounds on the delays of messages from the sender; **rtt** stores the last computed round trip time value; **period** denotes the period (e.g., five seconds); and **receiver** denotes the receiver to which it wants to know the round trip time. A **Receiver** object only has attributes bounding the delays of messages *from* the receiver.

A “delayed” message  $\text{dly}(m, t, t')$  denotes a message  $m$  whose *remaining* delay is in the interval  $[t, t']$ . The rule **deliver** removes the **dly** wrapper, thereby making the message “ripe,” whenever the lowest remaining delay has reached 0.

When the **Sender**’s **timer** expires (i.e., becomes 0), a new round of the RTT-finding protocol starts (rule **send**). The **Sender** sends an **rttReq** message with its current clock value  $T$  to the **Receiver**, with appropriate delay bounds. The timer is also reset to expire when the *next* iteration should start.

When a **Receiver** receives such a request, it replies with an **rttResp** message with the received timestamp  $T$ , with appropriate delay bounds (rule **respond**).

When the **Sender** receives this response, with its original timestamp  $T$ , it can easily compute and store the (latest) round trip time (rule **recordRTT**).

The tick rule in this system, which could have many **Sender/Receiver** pairs, is the usual one for object-oriented Real-Time Maude specifications [31]: time can advance in a state **STATE** by any amount less than or equal to  $\text{mte}(\text{STATE})$ , and the function **timeEffect** models how the passage of time affects the state. **mte** ensures that time cannot pass beyond the time when a message *must* be delivered, that time cannot pass when there is a “ripe” (un-delayed) message in the state, and that time cannot pass beyond the expiration time of any **timer**. **timeEffect** reduces the remaining bounds of all message delays and timer values, and increases the **clock** values, according to the elapsed time.

Finally, **init** defines an initial state with one **Sender** and one **Receiver** and the given lower and upper bounds on the delays sent from each object, so that each recorded RTT value should be between 12 and 50.

## 4 Analysis Using Maude’s Strategy Language Directly

This section explains how to perform “Real-Time Maude-style” time-sampling-strategy-based (both time-bounded and “clock-less” unbounded) reachability

analyses using Maude’s strategy language, instead of having to use the Real-Time Maude tool.

Assuming that the tick rule(s) are as described above, let *str-max* be the following strategy expression:

```
all |
  (matchrew CS:ClockedSystem
    such that {STATE} in time T2 := CS:ClockedSystem /\ mte(STATE) /= 0
    by CS:ClockedSystem
    using tick[T <- if mte(STATE) == INF then 4 else mte(STATE) fi])
```

*str-max* denotes the application of any executable rewrite rule (*all*) or a tick rule. If it is a tick rule, then the variable T denoting the time increase is instantiated to *mte(STATE)*, for the given STATE obtained using *matchrew*, unless *mte(INF)* equals INF, in which case T is set to 4. The tick rule is not applied if *mte(STATE)* is 0. Therefore, this expression denotes a single application of any rule under the *maximal* time sampling strategy (with default step 4).

*Unbounded search* for *n* (clocked) states matching a *state pattern pattern*, with maximal time sampling (*str-max* above), can be performed using the command:

```
srew [n] init using str-max * ; (match pattern in time T3:Time) .
```

*Example 1.* The following command checks whether it is possible to use *maximal* time sampling to find an RTT value 20:

```
Maude> srew [1] init using
  str-max * ; (match {< snd : Sender | rtt : 20, ATTS:AttributeSet >
    C:Configuration} in time T3:Time) .
```

Since the *clock* value of the sender can grow beyond any bound, and since the desired state is not reachable with the selected time sampling strategy, this command does not terminate. If we instead search for (two) reachable states where the (maximal) RTT value 50 has been recorded, we get answers:<sup>4</sup>

```
Maude> srew [1] init using
  str-max * ; (match {< snd : Sender | rtt : 50, ATTS:AttributeSet >
    C:Configuration} in time T3:Time) .
```

```
Solution 1
rewrites: 589 in 0ms cpu (0ms real) (792732 rewrites/second)
result ClockedSystem:
  {< snd : Sender | clock : 50, timer : 4950, lowerDly : 5,
    upperDly : 20, rtt : 50, receiver : rcv, period : 5000 >
  < rcv : Receiver | lowerDly : 7, upperDly : 30 >} in time 50
```

```
Solution 2
rewrites: 710 in 0ms cpu (0ms real) (734229 rewrites/second)
result ClockedSystem:
{< snd : Sender | rtt : 50, ... > < rcv : Receiver | ... >} in time 5000
```

*Deterministic* time sampling is easier: with such time sampling with step 1, states with RTT value 20 are indeed reachable:

<sup>4</sup> Parts of Maude code and Maude output will be replaced by ‘...’.

```
Maude> srew [2] init using
      (all | tick[T <- 1]) *
      ; (match {< snd : Sender | rtt : 20, ATTS:AttributeSet >
          C:Configuration} in time T3:Time) .
```

```
Solution 1
rewrites: 18450 in 16ms cpu (16ms real) (1107310 rewrites/second)
result ClockedSystem:
  {< snd : Sender | clock : 20, timer : 4980, rtt : 20, ... >
   < rcv : Receiver | ... >} in time 20
```

```
Solution 2
rewrites: 20409 in 18ms cpu (18ms real) (1102295 rewrites/second)
result ClockedSystem:
  {< snd : Sender | clock : 21, timer : 4979, rtt : 20, ... >
   < rcv : Receiver | ... >} in time 21
```

*Time-bounded Reachability Analysis.* We can perform *time-bounded* reachability analysis to find desired states reachable in a time interval  $[lower, upper]$  by: (i) applying tick rules only if the “system clock” does not go beyond *upper*, and (ii) only searching for state patterns of clocked states whose clocks are greater than or equal to *lower*. Such time-bounded reachability analysis should terminate, since the tick rule is not applied when the time bound has been reached.

*Example 2.* We search for states where the desired RTT values can be found in the time interval  $[5000, 10000]$ . With maximal time sampling, we search for two states reachable in the desired interval where the recorded RTT value is 50:

```
Maude> srew [2] init using
      (all |
       (matchrew CS:ClockedSystem
        such that {STATE} in time T2 := CS:ClockedSystem /\ mte(STATE) /= 0
        /\ T2 + (if mte(STATE) == INF then 4 else mte(STATE) fi) <= 10000
        by CS:ClockedSystem
        using tick[T <- if mte(STATE) == INF then 4 else mte(STATE) fi])) *
      ; (match {< snd : Sender | rtt : 50, ATTS:AttributeSet >
          C:Configuration} in time T3:Time s.t. T3:Time >= 5000) .
```

```
Solution 1
rewrites: 724 in 1ms cpu (1ms real) (498279 rewrites/second)
result ClockedSystem:
  {< snd : Sender | clock : 5000, rtt : 50, ... >
   < rcv : Receiver | ... >} in time 5000
```

```
Solution 2
rewrites: 749 in 1ms cpu (1ms real) (450933 rewrites/second)
result ClockedSystem:
  {< snd : Sender | rtt : 50, ... > < rcv : Receiver | ... >
   dly(rttReq 5000 from snd to rcv, 5, 20)} in time 5000
```

The same time-bounded reachability analysis terminates (with `No solution`) if we instead search for a state with recorded RTT value 20.

The following time-bounded command searches for two reachable states with recorded RTT value 20 using deterministic time sampling (with increment 1):

```
Maude> srew [2] init using
  (all |
    (matchrew CS:ClockedSystem
      such that {STATE} in time T2 := CS:ClockedSystem
        /\ mte(STATE) /= 0 /\ T2 + 1 <= 10000
      by CS:ClockedSystem
        using tick[T <- 1])) *
  ; (match {< snd : Sender | rtt : 20, ATTS:AttributeSet >
      C:Configuration} in time T3:Time s.t. T3:Time >= 5000) .
```

Solution 1

```
rewrites: 15188595 in 20685ms cpu (20835ms real) (734246 rewrites/second)
result ClockedSystem:
  {< snd : Sender | rtt : 20, ... > < rcv : Receiver | ... >} in time 5000
```

Solution 2

```
rewrites: 15189532 in 20687ms cpu (20837ms real) (734226 rewrites/second)
result ClockedSystem: ... in time 5000
```

*Unbounded Reachability Analysis Without System Clocks.* Our states have the form  $\{state\}$  in time  $clock$ ; i.e., they include the “system clock” denoting how much time has passed in the system since the execution started. This clock is unnecessary for *unbounded* reachability analysis, but makes the reachable (clocked) state space infinite, even when the (clock-less) reachable state space is finite.

The form of our tick rules gives us clocked states. For unbounded and unclocked analysis we can either manually transform the tick rules into the corresponding clock-less tick rules

```
cr1 [tick] : {t} => {u} if T <= mte(t) /\ cond .
```

and remove “in time 0” from the initial states. However, modifying the original specification for this single command may be undesired.

Nevertheless, the shape of our tick rules mandates that something is done to work on clock-less states. We can extend our modules with either the equation

```
eq {STATE} in time T = {STATE} .
```

or with following rule, that removes the clock, and always apply this rule right after applying a tick rule:

```
r1 [removeClock] : {STATE} in time T => {STATE} [nonexec] .
```

We mark this rule as non-executable to allow us to control when it can be applied (e.g., it will not be applied by the strategy `all`).

*Example 3.* We can perform “clock-less” search for an RTT value 20 with maximal sampling by removing the global clock after each tick:

```
Maude> srew [1] init using
  (all |
    (matchrew GS:GlobalSystem
      such that {STATE} := GS:GlobalSystem /\ mte(STATE) /= 0
      by GS:GlobalSystem
        using tick[T <- if mte(STATE) == INF then 4 else mte(STATE) fi] ;
        removeClock)) *
  ; (match {< snd : Sender | rtt : 20, ATTS:AttributeSet > C:Configuration}) .
```

Unfortunately, even the clock-less reachable state space is infinite in our example, since the sender has its own unbounded clock. The same search for RTT value 50 finds the desired state.

*Simulation.* Time-bounded *simulation* of one behavior can be performed by taking one (**one**) of the normal forms (!) when the tick rule is not applied if it would advance the “system clock” beyond the time bound:

*Example 4.* We can simulate one behavior of our (original) RTT example up to time 10.000 as follows:

```
Maude> srew [1] init using
(all |
  (matchrew CS:ClockedSystem
    such that {STATE} in time T2 := CS:ClockedSystem /\ mte(STATE) /= 0
      /\ T2 + (if mte(STATE) == INF then 4 else mte(STATE) fi) <= 10000
    by CS:ClockedSystem
    using tick[T <- if mte(STATE) == INF then 4 else mte(STATE) fi])) ! .

Solution 1
rewrites: 1509 in 2ms cpu (2ms real) (614163 rewrites/second)
result ClockedSystem:
  {< snd : Sender | rtt : 50, ... > < rcv : Receiver | ... >
   dly(rttReq 10000 from snd to rcv, 5, 20)} in time 10000
```

## 5 A Strategy Language for Real-Time Rewrite Theories

Section 4 shows that unbounded and time-bounded reachability with both maximal and deterministic time sampling can be performed using Maude’s strategy language. However, even these simple analysis methods need somewhat hard-to-understand strategy expressions. Furthermore, as shown in Section 5.1, where we discuss strategies for real-time systems, we often need more complex strategies. How can the non-Maude-expert analyze her system with such strategies?

To address this question, in this section we define what we hope is a powerful yet intuitive timed strategy language for real-time rewrite theories. Our focus on strategies for the non-Maude-expert is also motivated by our recent work on providing formal analysis and parameter synthesis for parametric timed automata and parametric time Petri nets using Maude with SMT solving [3,4], where one of our main selling points is the ability to analyze the system with user-defined execution strategies. However, this point becomes moot if the timed automata/Petri net expert cannot define her strategies.

Our strategy language for real-time rewrite theories supports, e.g.,

- *separate* definitions of strategies for *discrete behaviors*, including the interplay between discrete actions and time advance, and *timed strategies*;
- *state-dependent time sampling* strategies and *conditional* discrete strategies;
- *history-dependent* strategies allowing, for instance, “counting” the number of times some states have been reached, manipulating such counters, and making strategies dependent of the values of these counters; and

- intuitive syntax for “Real-Time Maude commands” with user-defined strategies.

Section 5.1 discusses important execution strategies for both real-time systems in general and real-time rewrite theories, and Section 5.2 introduces our timed strategy language. Section 5.3 shows how expressions in our strategy language can be translated into expressions in Maude’s strategy language, thereby giving it a formal semantics. Sections 5.4 and 5.5 show how most Real-Time Maude analysis methods can be performed using our strategy language, and Section 5.6 compares the performance of our analysis commands with standard Maude search on the CASH scheduling algorithm. The executable Maude definition of our language and its semantics is available at [24].

### 5.1 Strategies for Real-Time Systems

Interesting execution strategies of timed systems in general include:

1. *Eagerness* of certain (or all) actions: time should not advance when such actions can be taken.
2. Advance time (or “idle”) by  $f(s)$  in all states (or locations)  $s$  belonging to a set of states  $S$ , advance time by  $g(s')$  in all states  $s'$  belonging to the set of states  $S'$ , and so on.
3. Do not perform action  $a$  more/less than  $x$  times.
4. Always execute action  $a_i$  before action  $a_j$  when both are enabled.

These examples indicate that we can consider three “types” of strategies: (i) strategies on the “discrete behaviors” (such as items 3 and 4 above); (ii) strategies on how much to advance time (item 2); and (iii) combining these (item 1). This means that the user may want to specify a strategy restricting the discrete behaviors of a system, as well as a strategy for how to advance time. Therefore, we must be able to *compose* any *discrete strategy* with any *timed strategy*.

In Real-Time Maude, the selected (deterministic or maximal) time sampling strategy is used in all tick rule applications. However, with maximal time sampling we may miss too many behaviors, whereas with deterministic time sampling we may cover all possible behaviors (for discrete time), but at the cost of “visiting” each time point, even when the system is just “idling,” leading to *very* inefficient analyses. In our RTT example, the only behaviors covered using maximal time sampling are those where the recorded RTT value is always 50. Always advancing time by 1 covers all possible behaviors, but requires visiting 5000 time points in each period, even though less than 50 time points are interesting.

An efficient time sampling strategy that covers all (interesting) behaviors for discrete time is the following instance of item (2) above:

- increment time by 1 when an action *could* happen (in the next time instant);
- increment time maximally otherwise.

In the RTT system, we should increment time by 1 when there is a delayed message in the state<sup>5</sup>, and maximally when there is no message in the state (and the system is just idling until the next period begins). This suggests an efficient time sampling strategy for a large class of distributed real-time systems [19].

## 5.2 Our Timed Strategy Language

A strategy  $\langle \mu, \tau \rangle$  (of sort `UStrat`) in our timed strategy language consists of a user-defined *discrete strategy*  $\mu$  (of sort `UDStrat`), controlling the way instantaneous rules are applied and their interaction with time passage, and a *timed strategy*  $\tau$  (of sort `UTStrat`) defining a time sampling strategy:

```
sorts UStrat UTStrat UDStrat .
op <_,_> : UDStrat UTStrat -> UStrat .
```

The discrete strategy  $\mu$  controls whether some (and if so, which) action/instantaneous rule must be applied in the current state, or whether some tick rule must be applied. The timed strategy  $\tau$  defines exactly how each “tick rule application” (i.e., each `delay` step) in the discrete strategy  $\mu$  is applied.

We extend the global state of the system with a map that stores information about the execution history. This allows us to define *history-dependent strategies*; i.e., strategies that depend on the current and the previously visited states:

```
sort StrState .
pr MAP{K, V} * (sort Entry{K, V} to Entry, sort Map{K, V} to Map) .

op |_| : ClockedSystem Map -> StrState .
```

The sorts `K` and `V` for the keys and their values are user-defined.

*Discrete Strategies.* Discrete strategies are defined using a language whose syntax is given as follows.

```
--- Intervals
sort Interval .
op [_,_] : Time Time -> Interval .
--- Conditions
sort SCond .
op matches_s.t._      : ClockedSystem Bool      -> SCond .
op matches_s.t._      : StrState Bool           -> SCond .
op matches_s.t._      : Map Bool                -> SCond .
op matches             : ClockedSystem           -> SCond .
op in_                 : Interval                -> SCond .
ops after before after= before= : Time          -> SCond .
ops _/\_ _\/_         : SCond SCond             -> SCond .
op not_                : SCond                  -> SCond .
--- User-defined strategies
op apply_              : Qid                     -> UDStrat .
ops apply[_] eager[_]  : QidList                 -> UDStrat .
ops action delay eager :                      -> UDStrat .
ops _;_ _or_ _or-else_ : UDStrat UDStrat          -> UDStrat .
op if_then_else_      : SCond UDStrat UDStrat    -> UDStrat .
ops stop skip         :                      -> UDStrat .
op get_and set_       : Map Map                  -> UDStrat .
```

<sup>5</sup> A further optimization would advance time to when the least remaining delay is 0.

Terms of sort `SCond` define conditions in some of the strategies. The condition `matches P s.t. C`, where  $P$  is a pattern and  $C$  is an (optional) boolean condition, checks whether the current state matches  $P$  so that  $C$  holds in the state. The pattern  $P$  can be a `ClockedSystem`, or a `StrState` (a clocked system extended with a `Map`). Other basic conditions include checking whether the current value  $t$  of the global clock satisfies:  $t \in [a, b]$  (`in [a, b]`),  $t > t'$  (`after t'`),  $t \geq t'$  (`after= t'`),  $t < t'$  (`before t'`), and  $t \leq t'$  (`before= t'`). Larger conditions can be constructed using conjunction, disjunction, and negation.

User-defined discrete strategies are: `apply ℓ` applies the instantaneous rule with label  $\ell$  *once*; `apply [ℒ]` applies *once* the first rule in the list of labels  $\mathcal{L}$  that succeeds in the current state (i.e.,  $\mathcal{L}$  defines a *priority* on the next rule to be applied); `action` applies *any* instantaneous rule *once*; `delay` applies a tick rule *once*; `eager` applies the instantaneous rules as much as possible, followed by *one* “delay” when it is possible; `eager [ℒ]` applies as much as possible the rules in the list  $\mathcal{L}$  followed by one “delay”;  $\mu ; \mu'$  is the sequential composition of two strategies;  $\mu$  or  $\mu'$  returns the union of the results obtained from the strategies  $\mu$  and  $\mu'$ ;  $\mu$  `or-else`  $\mu'$  applies  $\mu$ , but applies the strategy  $\mu'$  if  $\mu$  fails; `if  $\phi$  then  $\mu$  else  $\mu'$`  is the conditional strategy; `stop` is the strategy that always fails; `skip` leaves the current state unchanged; and `get M and set M'` uses the pattern  $M$  to retrieve (part of) the map storing information about the execution of the strategy and updates it according to  $M'$ .

*Time Sampling Strategies.* Time sampling strategies are defined as follows:

```
op fixed-time_      : Time          -> UTStrat .
op max-time with default_ : Time      -> UTStrat .
sorts CTStrat LCTStrat . --- Time sampling strategies defined by cases
subsort CTStrat < LCTStrat .
op when_do_        : SCond UTStrat   -> CTStrat .
op switch_otherwise_ : LCTStrat UTStrat -> UTStrat .
```

`fixed-time t` advances the time by time  $t$  in each application of a tick rule (where advancing time by that amount is possible). `max-time with default t` advances time in a tick rule application by the maximal time  $t'$  possible *for that tick rule*, and advances time by  $t$  if  $t'$  is `INF`. The conditional time sampling strategy `switch cases otherwise  $\tau$` , where *cases* is a list of choices of the form `when  $\phi_j$  do  $\tau_j$` , executes the first strategy  $\tau_i$  whose guard  $\phi_i$  holds in the current state; the strategy  $\tau$  is applied if none of the guards hold.

*Example 5.* Consider the RTT system in Section 3.2. A basic execution strategy, for any timed strategy  $\tau$ , applies *any* enabled (instantaneous or tick) rule *once*:

```
< delay or action ,  $\tau$  >
```

It is also possible to analyze the system by assuming that instantaneous rules have a higher priority than the tick rule:

```
< eager ;  $\tau$  >
```

In fact, we can give preference to the rules `send` and `respond`, then to the other actions, and finally to the tick rule:



```
< (apply ['send 'respond] or-else action or-else tick,  $\tau$  >
```

Regarding the time sampling strategy, for any discrete strategy  $\mu$ , the best choice for this system is: if there is a *delayed* message in the state, increase time by 1, otherwise increase time maximally. This *state-dependent* time sampling strategy can be defined as follows:

```
<  $\mu$  , switch when matches ({CONF dly(M, T1, T2)} in time R) do fixed-time 1
  otherwise max-time with default 1 >
```

When the network is working well, we could save bandwidth by not performing the RTT-finding procedure in *each* period. We therefore add the following rewrite rule, that allows a **Sender** to skip a round of the protocol by just resetting the timer when it expires (instead of also sending an `rttReq` message):

```
omod RTT-WITH-IDLING is including RTT .
var S : Oid . vars T T2 : Time .

rl [skipRound] :
  < S : Sender | timer : 0, period : T2 > => < S : Sender | timer : T2 > .
endom
```

When its timer expires, a sender therefore nondeterministically chooses between executing a round of the protocol (rule `send`) or skipping one round (rule `skipRound`).

A sensible strategy is to skip some rounds but never skip more than two rounds in a row. To define this *state- and history-dependent* strategy, we use a counter labeled with 'C' to avoid skipping “more than two rounds”:

```
< delay or
  if matches {< S : Sender | timer : 0, ATTS >} in time T --- State dep.
  then if (matches ('C |-> N) s.t. N <= 1) --- History dependent
    then apply 'skipRound ;
      (get ('C |-> N) and set ('C |-> N + 1)) --- Skip and increment
    else apply 'send ;
      (get ('C |-> N) and set ('C |-> 0)) --- Send and reset
  else action ) ,  $\tau$  >
```

### 5.3 Semantics

This section shows how expressions in our timed strategy language can be translated into expressions in Maude’s strategy language. Hence, the denotational and operational semantics of the latter [15] formally describes the execution of real-time rewrite theories controlled by a timed strategy  $\langle \mu, \tau \rangle$ .

We define a map  $\llbracket - \rrbracket$  from terms of sort **UStrat** to terms of sort **Strategy**, the sort in Maude’s prelude used to meta-represent strategies.

**Definition 1 (Semantics).** *The interpretation of conditions  $\llbracket - \rrbracket_b$ , time sampling strategies  $\llbracket - \rrbracket_t$ , and real-time strategies  $\llbracket - \rrbracket$ , as terms of sort **Strategy** is given in Fig. 1. These definitions use the following variables, and require the new operator and rule below:*

```
vars M M' M'' : Map . var CS : ClockedSystem . var SS : StrState .
```

$$\begin{aligned}
\llbracket \text{matches } SS \text{ s.t. } B \rrbracket_b &= \llbracket \text{matches } \overline{SS} \text{ s.t. } \overline{B} \rrbracket_b \\
\llbracket \text{matches } CS \text{ s.t. } B \rrbracket_b &= \llbracket \text{matches } CS \mid M \text{ s.t. } \overline{B} \rrbracket_b \\
\llbracket \text{matches } M \text{ s.t. } B \rrbracket_b &= \llbracket \text{matches } \overline{CS} \mid (M, M') \text{ s.t. } \overline{B} \rrbracket_b \\
\llbracket \text{matches } Te \text{ s.t. } Te' \rrbracket_b &= \text{match } Te \text{ s.t. } Te' \\
\llbracket \text{after}(T) \rrbracket_b &= \llbracket \text{match } \{ CS \} \text{ in time } T' \text{ s.t. } T > T' \rrbracket_b \\
\llbracket \phi_1 \wedge \phi_2 \rrbracket_b &= \llbracket \phi_1 \rrbracket_b ; \llbracket \phi_2 \rrbracket_b \quad \llbracket \phi_1 \vee \phi_2 \rrbracket_b = \llbracket \phi_1 \rrbracket_b \text{ or-else } \llbracket \phi_2 \rrbracket_b \\
\llbracket \text{not } \phi \rrbracket_b &= \text{not } \llbracket \phi \rrbracket_b
\end{aligned}$$

(a) Conditions. Definitions for `before`, `in`, etc are similar and omitted.

$$\begin{aligned}
\llbracket \text{fixed-time } T1 \rrbracket_t &= \text{'tick } [\overline{T} \leftarrow \overline{T1}] \{ \text{empty} \} \\
\llbracket \text{max-time with default } T1 \rrbracket_t &= \text{matchrew } \overline{SS} \text{ s.t. } \overline{\{ S \} \text{ in time } T2 \mid M} := \overline{SS} \\
&\text{ by } \overline{SS} \text{ using 'tick } [\overline{T} \leftarrow \text{if } INF == \text{mte}(S) \text{ then } T1 \text{ else } \text{mte}(S) \text{ fi}] \{ \text{empty} \} \\
\llbracket \text{switch (when } C \text{ do } \tau \text{) LC otherwise } \tau' \rrbracket_t &= \llbracket C \rrbracket_b ? \llbracket \tau \rrbracket_t : \llbracket \text{switch LC otherwise } \tau' \rrbracket_t \\
\llbracket \text{switch (when } C \text{ do } \tau \text{) otherwise } \tau' \rrbracket_t &= \llbracket C \rrbracket_b ? \llbracket \tau \rrbracket_t : \llbracket \tau' \rrbracket_b
\end{aligned}$$

(b) Timed strategies.

$$\begin{aligned}
\llbracket \langle \text{stop}, \tau \rangle \rrbracket &= \text{fail} \quad \llbracket \langle \text{skip}, \tau \rangle \rrbracket = \text{idle} \quad \llbracket \langle \text{apply } Q, \tau \rangle \rrbracket = Q \text{ [none] } \{ \text{empty} \} \\
\llbracket \langle \text{action}, \tau \rangle \rrbracket &= \text{all} \quad \llbracket \langle \text{delay}, \tau \rangle \rrbracket = \llbracket \tau \rrbracket_t \quad \llbracket \langle \text{eager}, \tau \rangle \rrbracket = \text{all} ! ; \text{try}(\llbracket \tau \rrbracket_t) \\
\llbracket \langle \text{apply } [\text{nil}], \tau \rangle \rrbracket &= \text{fail} \quad \llbracket \langle \text{apply } [Q \text{ LQ}], \tau \rangle \rrbracket = \text{apply } Q \text{ or-else } \llbracket \text{apply } [LQ] \rrbracket \\
\llbracket \langle \text{eager } [L], \tau \rangle \rrbracket &= \llbracket \langle \text{apply } [L], \tau \rangle \rrbracket ! : \text{try}(\llbracket \tau \rrbracket_t) \\
\llbracket \langle \mu ; \mu', \tau \rangle \rrbracket &= \llbracket \langle \mu, \tau \rangle \rrbracket ; \llbracket \langle \mu', \tau \rangle \rrbracket \quad \llbracket \langle \mu \text{ or } \mu', \tau \rangle \rrbracket = \llbracket \langle \mu, \tau \rangle \rrbracket \mid \llbracket \langle \mu', \tau \rangle \rrbracket \\
\llbracket \langle \mu \text{ or-else } \mu', \tau \rangle \rrbracket &= \llbracket \langle \mu, \tau \rangle \rrbracket \text{ or-else } \llbracket \langle \mu', \tau \rangle \rrbracket \\
\llbracket \langle \text{if } C \text{ then } \mu \text{ else } \mu', \tau \rangle \rrbracket &= \llbracket C \rrbracket_b ? \llbracket \langle \mu, \tau \rangle \rrbracket : \llbracket \langle \mu', \tau \rangle \rrbracket \\
\llbracket \langle \text{get } M' \text{ and set } M'', \tau \rangle \rrbracket &= \text{matchrew } \overline{SS} \text{ s.t. } \overline{\{ S \} \text{ in time } T1 \mid (M, M')} := \overline{SS} \\
&\text{ by } \overline{SS} \text{ using 'updateMap } [\overline{M} \leftarrow \overline{M} ; \overline{M'} \leftarrow \overline{M'} ; \overline{M''} \leftarrow \overline{M''}] \{ \text{empty} \}
\end{aligned}$$

(c) Discrete and real-time strategies.

$$\begin{aligned}
\llbracket \langle \text{check } \phi, \tau \rangle \rrbracket &= \llbracket \phi \rrbracket_b \quad \llbracket \langle \text{until } \phi \text{ do } \mu, \tau \rangle \rrbracket = (\llbracket \phi \rrbracket_b ? \text{fail} : \llbracket \langle \mu, \tau \rangle \rrbracket) ! \\
\llbracket \langle \text{repeat } \mu, \tau \rangle \rrbracket &= \llbracket \langle \mu, \tau \rangle \rrbracket * \quad \llbracket \langle 0 \text{ steps with } \mu, \tau \rangle \rrbracket = \text{idle} \\
\llbracket \langle \text{s}(N) \text{ steps with } \mu, \tau \rangle \rrbracket &= \llbracket \langle \mu, \tau \rangle \rrbracket ; \langle N \text{ steps with } \mu, \tau \rangle \\
\llbracket \text{untime } \tau \rrbracket_t &= \llbracket \tau \rrbracket_t ; \text{'removeClock } [\text{none}] \{ \text{empty} \}
\end{aligned}$$

(d) General timed strategies.

Fig. 1: Interpretation of real-time strategies as Maude's strategies.

```

var B : Bool . vars Te Te' : Term . vars T T' T1 T2 : Time .
var C : SCond . var LC : LCTStrat . var S : System .

op matching_s.t._ : Term Term -> SCond .
rl [updateMap] : CS | (M, M') => CS | (M, M'') [nonexec] .
    
```

In Figure 1, we use  $\bar{t}$  to denote the meta-representation of a term  $t$  (in Maude,  $\text{upTerm}(t)$ ). For instance, the second case in Figure 1a must be read as

$$\llbracket \text{matching CS s.t. B} \rrbracket_b = \llbracket \text{matching '}_|_|\_[\overline{\text{CS}}, \text{'M:Map}] \text{ s.t. } \overline{\text{B}} \rrbracket_b$$

and specified in Maude as

```

eq enc(matching CS s.t. B) =
  enc(matching '}_|_|\_[upTerm(CS), 'M:Map] s.t. upTerm(B))
    
```

The Maude strategy  $\llbracket \phi \rrbracket_b$  fails when condition  $\phi$  does not hold, and succeeds (without modifying the current state) otherwise. `matches` expressions are reduced until their parameters are of sort `Term`. Then, Maude's strategy `match` is used to check whether the current state matches the pattern and satisfies the given condition (otherwise, `match` fails).

The Maude strategy  $\llbracket \tau \rrbracket_t$  applies the tick rule by instantiating the variable `T` with the needed substitution according to  $\tau$ . In `max-time`, Maude's strategy `matchrew` is used to do pattern matching and bind the variable `S` with the current configuration. Hence, the call `mte(S)` determines the next tick value. The definition of `switch` uses the conditional Maude strategy  $\alpha ? \beta : \gamma$  to choose the right time sampling strategy  $\tau_i$ .

The Maude strategy  $\llbracket \langle \mu, \tau \rangle \rrbracket$  fails when  $\mu = \text{stop}$  and does nothing if  $\mu = \text{skip}$ . If  $\mu = \text{apply Q}$ , the rule with label `Q` is applied, without any substitution (`[none]`) and with the `{empty}` list of strategies (since no particular strategy is used to solve rewrite expressions in conditional rules). Maude's strategy `all` non-deterministically chooses, and applies once, any of the *executable* rewrite rules. Therefore, when  $\mu = \text{action}$ , only executable instantaneous (and no tick) rules are applied. The strategy  $\llbracket \text{apply } [\mathcal{L}] \rrbracket$  tries, in order, the instantaneous rules in the list  $\mathcal{L}$ . In the case  $\mu = \text{delay}$ , the strategy  $\llbracket \tau \rrbracket_t$  is executed. The normalization operator `all !` applies `all` until it cannot be further applied. Hence, when  $\mu = \text{eager}$ , all the instantaneous transitions are (non-deterministically) taken and then, if it is possible, a tick is performed. The interpretation of the strategies `_;_`, `_or_`, `_or-else_` and `if-then-else` uses the corresponding constructors in Maude's strategy language. In the case  $\mu = \text{get M' and set M''}$ , Maude's `matchrew` is used to bind  $M'$  with the needed entries in the map storing information about the execution of the strategy. Then, the execution of the rule `updateMap` replaces the values in  $M'$  with the corresponding ones in  $M''$ .

The “extended” discrete and timed strategies implementing the different analyses proposed in Section 5.4 are defined as Maude's strategies as shown in Figure 1d. The Maude's strategy  $\llbracket \langle \text{check } \phi, \tau \rangle \rrbracket$  executes  $\llbracket \phi \rrbracket_b$  (thus failing when  $\phi$  does not hold). The Maude's strategy  $\llbracket \langle \text{until } \phi \text{ do } \mu, \tau \rangle \rrbracket$  uses the normalization operator to repeatedly execute  $\llbracket \langle \mu, \tau \rangle \rrbracket$  until the point where  $\llbracket \phi \rrbracket_b$  succeeds. The

iteration operator  $*$  is used to give meaning to the strategy `repeat  $\mu$` , that iteratively executes  $\mu$  until it fails. The Maude’s strategy  $\llbracket \langle n \text{ steps with } \mu, \tau \rangle \rrbracket$  stops executing  $\mu$  when  $n = 0$ . Otherwise, it is recursively called after applying  $\llbracket \langle \mu, \tau \rangle \rrbracket$ . Finally, the strategy  $\llbracket \text{untime } \tau \rrbracket_t$  executes  $\llbracket \tau \rrbracket_t$  and then applies the rule `removeClock`.

#### 5.4 User-Friendly Analysis Commands

A user-defined strategy  $\langle \mu, \tau \rangle$  controls “one round” of the execution of the system. In this section we provide convenient “Real-Time Maude-like” syntax for most simulation, reachability and other formal analysis methods provided by Real-Time Maude, albeit executed with user-defined strategies.

We define new discrete and time sampling strategies (sorts `DStrat`, `TStrat` and `Strat`) that control how user’s strategies are applied. Building on these new strategies, different analysis for real-time rewrite theories can be neatly defined.

Discrete strategies (of sort `DStrat`), besides the basic user-defined discrete strategies, include: the strategy `check  $\phi$`  that fails if  $\phi$  does not hold in the current state; the conditional repetition of a given strategy `until  $\phi$  do  $\mu$` ; the strategy `repeat  $\mu$`  that iteratively executes  $\mu$  until it fails; and the strategy  `$n$  steps with  $\mu$`  that repeats  $n$  times  $\mu$ . Overloaded operators for the sort `DStrat` (e.g., `op _;_ ... [ditto]`) are also defined and omitted here.

```
sort DStrat .      subsort UDStrat < DStrat .
op check_         : SCond                -> DStrat .
op until_do_     : SCond DStrat          -> DStrat .
op repeat_       : DStrat                -> DStrat .
op _steps with_  : Nat DStrat            -> DStrat .
```

General timed strategies (of sort `TStrat`) extend user-defined time sampling strategies with a new case, used later to define untimed reachability analysis:

```
sort TStrat .      subsort UTStrat < TStrat .
op untime        : TStrat -> TStrat .
```

`untime  $\tau$`  applies  $\tau$  and then the rule `removeClock`, thus removing the global clock from the current state. These new strategy constructors are defined as Maude strategies as shown in Figure 1d.

*Commands.* We now define a convenient syntax for most Real-Time Maude-like analysis commands using strategies. Given a user-defined strategy  $\langle \mu, \tau \rangle$ , we define an extended strategy  $\langle \mu', \tau' \rangle$  that implements such an analysis command by rewriting (using Maude’s `metaSrewrite`) an initial state `init` and returning a list of `ClockedSystems` (the solutions).

Time-bounded simulation is implemented with the following command:

```
op tsim [_] in_:_using_with sampling_until_ :
  Nat Qid StrState DStrat TStrat Time -> LClockedSystem .
```

The command `tsim [n] in R : init using  $\mu$  with sampling  $\tau$  until r` returns the first  $n$  states that result when rewriting `init` in theory `R` when following the strategy  $\langle \langle \text{until after}=(r) \text{ do } \mu \rangle, \tau \rangle$ . In words, this strategy returns the

first reachable states, in each branch of the search tree, whose global clock  $t$  satisfies  $t \geq r$ . The parameter `[n]` is optional.

It is also possible to observe the behavior of the system up to a given number  $d$  of rewriting steps:

```
op trew [_,_] in_:using_with sampling_ :
  Nat Nat Qid StrState DStrat TStrat -> LClockedSystem .
```

The command `trew [d, n] in R : init using  $\mu$  with sampling  $\tau$`  returns the first  $n$  states that can be reached after  $d$  rewriting steps. For that, the initial state is rewritten with the strategy `(d steps with  $\mu$ ),  $\tau$` .

Unbounded and time-bounded reachability analyses are implemented by the following commands:

```
op tsearch [_] in_:=>_using_with sampling_ :
  Nat Qid StrState SCond DStrat TStrat -> LClockedSystem .
op tsearch [_] in_:=>_using_with sampling_in time_ : ... -> ... .
```

`tsearch [n] in R : init =>  $\phi$  using  $\mu$  with sampling  $\tau$`  returns the first  $n$  states that result from `init` by rewriting with the strategy `(repeat  $\mu$  ; check  $\phi, \tau$ )`, i.e., repeat  $\mu$  zero or more times and, on the resulting term, check  $\phi$ . Time-bounded reachability analysis `tsearch [n] in R : init =>  $\phi$  using  $\mu$  with sampling  $\tau$  in time [a,b]` is implemented as the extended strategy `(repeat (if after(b) then stop else  $\mu$ ) ; check ( $\phi \wedge$  in [a,b]),  $\tau$ )`.

“Depth-bounded” versions of the form `tsearch [n,d] in R...` of the above commands are available; they check whether a  $\phi$ -state can be reached by applying at most  $d$  times the strategy  $\mu$  (`d steps using  $\mu$  | d-1 steps using  $\mu \dots$` ). Furthermore, similar commands `dsearch` are defined where `metaSrewrite` is invoked with the flag `depthFirst`, thus exploring the rewriting graph in depth.

Untimed reachability analysis is possible with the command

```
op usearch [_] in_:=>_using_with sampling_ :
  Nat Qid StrState SCond DStrat TStrat -> LClockedSystem .
```

The implementation of this command is similar to the one for `tsearch` but the sampling strategy used is `untime( $\tau$ )`: after each tick, the global clock is removed from the state. A depth-first version `dusearch` is also available.

Finding the longest and the shortest time it takes to reach a desired state is supported by the following commands:

```
op find latest in_:=>_using_with sampling_ :
  Qid StrState SCond DStrat TStrat -> LClockedSystem .
op find earliest in_:=>_using_with sampling_ : ... -> LClockedSystem .
```

`find latest` uses `metaSrewrite` to find *all* the solutions when rewriting the initial state with the strategy `(until  $\phi$  do  $\mu$  ; check  $\phi, \tau$ )`. This finds the *first* state in *all* the branches of the search tree that satisfies  $\phi$ . We then post-process the returned list to find the state with the greatest global clock value. This command may not terminate if there is a branch where  $\phi$  never holds.

The command `find earliest` cannot be implemented with a procedure similar to the one for `find latest`. The reason is that `find earliest` must finish if there is a reachable state satisfying  $\phi$ , even if there are branches not leading

to  $\phi$ -states. We therefore compute the first solution when applying the strategy  $\langle \text{until } \phi \text{ do } \mu ; \text{check } \phi , \tau \rangle$ . Let  $t$  be the global clock value in this state. The strategy  $\langle \text{until } \psi \text{ do (if after}(t) \text{ then stop else } \mu) ; \text{check } \psi , \tau \rangle$ , where  $\psi$  is the condition  $\phi \wedge \text{before}(t)$ , is then used to find a new solution (if any) whose global clock is strictly smaller than  $t$ . This branch-and-bound procedure is repeated until no further solutions are found.

### 5.5 Example: Analyzing the Round Trip Time Protocol

We illustrate the use of our timed strategy language on the RTT example.

We check whether an RTT value 20 can be recorded when incrementing time by 1 in each tick rule application:<sup>6</sup>

```
Maude > red tsearch [2] in 'RTT : init =>
      matches ({CONF < S : Sender | rtt : 20, ATTS >} in time R:Time)
      using delay or action with sampling fixed-time 1 .

result NeList{ClockedSystem}:
  ({< snd : Sender | rtt : 20, ... > < rcv : Receiver | ... >} in time 20)
  ({< snd : Sender | rtt : 20, ... > < rcv : Receiver | ... >} in time 21)
```

It is possible to reach two states with RTT 50 using maximal time sampling:

```
Maude > red tsearch [2] in 'RTT : init =>
      matches ({CONF < S : Sender | rtt : 50, ATTS > } in time R:Time )
      using delay or action with sampling max-time with default 4 .

result NeList{ClockedSystem}:
  ({< snd : Sender | clock : 50, timer : 4950, rtt : 50, ... >
   < rcv : Receiver | lowerDly : 7, upperDly : 30 >} in time 50)
  ({< snd : Sender | clock : 5000, rtt : 50, ... >
   < rcv : Receiver | lowerDly : 7, upperDly : 30 >} in time 5000)
```

As already shown in Example 1, it is not possible to find an RTT value 20 when the maximal time-sampling strategy is used. Consider the *state-based* sampling strategy in Example 5, where time advances by 1 when there are delayed messages, and maximally otherwise:

```
< action or delay ,
  switch when matches ({CONF dly(M, T1, T2)} in time R) do fixed-time 1
  otherwise max-time with default 1 >
```

This strategy allows us to find states with RTT value 20 while visiting less states when compared to the *fixed-time* strategy. Note, for instance, the global clock of the second solution found by the command below, and compare it with the outputs in the two commands above:

```
Maude> tsearch [2] in 'RTT : init => matches ...
      using delay or action with sampling  $\tau$  .

result NeList{ClockedSystem}:
  ({< snd : Sender | rtt : 20, ... > ... } in time 20)
```

<sup>6</sup> Symbols in all capitals are variables, whose declarations we often omit.

```
{< snd : Sender | rtt : 20, ... > ... } in time 5000)
```

We can also search for a state with RTT value 50 reachable in the time interval [5000,10000] with maximal time sampling:

```
Maude> red tsearch [2] in 'RTT : init => matches ... using delay or action
      with sampling max-time with default 4 in time [5000, 10000] .
```

```
result NeList{ClockedSystem}:
  ({< snd : Sender | rtt : 50, ... > } in time 5000)
  ({< snd : Sender | rtt : 50, ... > dly(...)} in time 5000)
```

The same search for an RTT value 20 yields no solution:

```
Maude> red tsearch [2] in 'RTT : init => matches ... using delay or action
      with sampling max-time with default 4 in time [5000, 10000] .
```

```
result LClockedSystem: (nil).LClockedSystem --- (No solution)
```

Untimed and “clock-less” search for RTT value 50:

```
Maude> red usearch [2] in 'RTT : init => matches ...
      using delay or action with sampling max-time with default 4 .
```

```
result NeList{ClockedSystem}:
  {< snd : Sender | rtt : 50, ... > < rcv : Receiver | ... >}
  {< snd : Sender | rtt : 50, ... > < rcv : Receiver | ... >}
```

Time-bounded simulation simulates one system behavior up to time 10000:

```
Maude> red tsim [1] in 'RTT : init
      using delay or action with sampling max-time with default 4 until 10000 .

result ClockedSystem: {< snd : Sender | rtt : 50, ... >
  < rcv : Receiver | lowerDly : 7, upperDly : 30 >}
  in time 10000
```

We then check the longest and shortest time needed to record an RTT value different from 0 and INF (`rtt?(STATE)`) for the first time in each behavior:

```
Maude> red find earliest in 'RTT : init => matches ({ STATE} in time T2)
      s.t. (rtt?(STATE)) using action or delay with sampling fixed-time 1 .
```

```
result ClockedSystem: {< snd : Sender | rtt : 12, ... > ... } in time 12
```

```
Maude> red find latest in 'RTT : ... with sampling fixed-time 1 .
```

```
result ClockedSystem: {< snd : Sender | rtt : 50, > ... } in time 50
```

```
Maude> red find latest ... using eager with sampling fixed-time 1 .
```

```
result ClockedSystem: {< snd : Sender | rtt : 12, ... > ... } in time 12
```

Finally, let  $\mu$  be the history-depend strategy in Example 5. The execution of the following commands show that some states are not visited when this strategy is followed, resulting in a smaller state space. (`size(L)` returns the size of the list L).

```
Maude> red size(tsearch in 'RTT : init => matches ({STATE} in time T2)
      using  $\mu$  with sampling max-time ... in time [0, 100000]) .
```

```

result NzNat: 126

Maude> red size(tsearch in 'RTT : init => matches ({STATE} in time T2)
        using action or delay with sampling max-time ... in time [0, 100000]) .

result NzNat: 162

```

## 5.6 Benchmarking

We have proposed and implemented a strategy language and shown that all (except for temporal logic model checking at the moment) Real-Time Maude analysis can be performed using strategies, with much more flexibility in defining both discrete strategies and time sampling strategies. The question is whether such Maude-strategy-based implementation of reachability analysis has competitive performance, compared to standard Maude reachability analysis.

We therefore compare the performance of our implementation with standard Maude search on a Maude model of a Real-Time Maude benchmark systems, a variation of the CASH scheduling algorithm developed by Marco Caccamo at UIUC [11]. The idea of the CASH is that some jobs may not need all the execution time allocated to it. These unused clock cycles are put in a *queue*, so that other jobs could use them to improve system performance. CASH is a sophisticated protocol, with sporadic tasks (i.e., a job could arrive at any time), unknown length of each job, and a queue of unused execution times.

Real-Time Maude analysis in [28] showed, somewhat surprisingly, that the length of the queue of unused execution times could grow beyond any bound. This means that most formal real-time tools cannot analyze this version of CASH. Real-Time Maude analysis discovered that a hard deadline could be missed [28]. This was a subtle (and previously unknown) flaw: the deadline miss was not found by extensive randomized simulation.

We transformed the Real-Time Maude specification of the CASH protocol into a “standard” Maude model by incrementing time by one unit in the tick rules (see [24]). We can therefore use Maude’s `search` command to find whether it is possible to reach a state where a deadline is missed within time 12:<sup>7</sup>

```

Maude> search [1] init =>* {DEADLINE-MISS CONF} in time T s.t. T <= 12 .

Solution 1 (state 599272)
rewrites: 34093729 in 14910ms cpu (14937ms real) ...
...

```

The `tsearch` and `dtsearch` (depth-first search) commands in our language that correspond to this time-bounded reachability query are executed as follows:

```

Maude> red tsearch [1] in 'CASH : init =>
        matches ({DEADLINE-MISS CONF} in time T))
        using delay or action with sampling fixed-time 1 in time [0, 12] .

```

<sup>7</sup> Here `init` denotes an initial state from which a missed deadline should not be reachable if the optimized version of CASH were correct.



```
rewrites: 44 in 19517ms cpu (19538ms real) ...
```

```
Maude> red dtsearch [1] ... in time [0, 12] .
```

```
rewrites: 44 in 2079ms cpu (2083ms real) ...
```

For a possibly fairer comparison, we also perform *unbounded* reachability analysis with Maude `search`, `tsearch`, and `usearch`; `dtsearch` failed to terminate in this case.

We first apply Maude’s `search` command, without constraints on the system clock:

```
Maude> search [1] init =>* {DEADLINE-MISS C:Configuration} in time T:Time .
```

```
Solution 1 (state 599272)
```

```
states: 599273 rewrites: 34093728 in 14988ms cpu (15010ms real) ...
```

And then compare it to our own unbounded `tsearch` command and our `usearch` command, which always applies a rule which removes the system clock after each tick step:

```
Maude> red tsearch [1] ... .
```

```
rewrites: 36 in 21906ms cpu (21951ms real) ...
```

```
Maude> red usearch [1] ... .
```

```
rewrites: 39 in 9778ms cpu (9790ms real) ...
```

For an even more optimized Maude `search`, we can modify our Maude specification by *manually* removing the “in time ...” part of each each tick rule, so that the state does not carry the system clock:

```
search [1] init =>* {DEADLINE-MISS CONF} .
```

```
Solution 1 (state 151069)
```

```
states: 151070 rewrites: 8601350 in 4498ms cpu (4503ms real) ...
```

All experiments were run on a Dell XPS 13 laptop (with an Intel i7 processor @ 1.30GHz and 16GB of RAM). For time-bounded reachability analysis, our `tsearch` command (20 seconds runtime) is not much slower than Maude’s `search` command (15 seconds) on a Maude model where the deterministic time sampling strategy with increment 1 is hard-coded in the tick rule. Furthermore, our “depth-first” search command `dtsearch` significantly outperforms Maude’s `search` command on this application (2 seconds).

In the unbounded case, it is fair to compare `tsearch` (22 seconds), which carries the system clock, to the Maude `search` which took 15 seconds, and `usearch` (9.8 seconds), which explicitly removes the clock after each tick, to Maude `search` of the manually modified model which never introduces the system clock (4.5 seconds).

To summarize, our strategy-implemented commands are not much more than twice as slow as Maude `search`, and in one case even significantly faster.

Because of the high degree of time-nondeterminism in the model (jobs can arrive at any time and may execute for an arbitrary amount of time), the only time sampling strategy that makes sense for CASH is deterministic with incre-

ment 1. The performance gain should be much larger on systems such as RTT where we should use mixed time sampling to “pass” idling states where not much can happen. Although these preliminary results are quite promising, we should perform more thorough benchmarking in future work.

## 6 Related Work

*Strategies for real-time systems.* UPPAAL STRATEGO [14] extends the timed automaton tool UPPAAL [7] with strategies and model checking under such strategies, where strategies are UPPAAL queries. UPPAAL STRATEGO seems mainly to be used in connection with machine learning-based synthesis of controller strategies. We target a more expressive model, provide a language for specifying actual strategies instead of queries, and must also provide time sampling strategies. We also perform model checking w.r.t. a strategy, but do not support synthesizing strategies.

*Rewrite strategies.* Different strategy languages have been proposed to cope with the inherent non-determinism in rewriting systems. Examples of such languages include ELAN [9], Stratego [10], and  $\rho$ Log [21]. Applications of Maude’s strategy language [15] include the analysis of different systems and models such as neural network [40] and membrane systems [38], as well as the specification of semantics of programming languages [17] and process calculi [35].

Our previous work [2,4,3] showed, for the first time, how Maude strategies can be used on simple real-time systems. That work motivated the development of an easy-to-use timed strategy language, which resulted in this paper.

Recently, Rubio et al. [36,37] have shown how to model check strategy-aware rewriting logic specification. These techniques have been implemented in the `umaudemc` tool, that allows for model checking LTL and CTL formulas, as well as to perform probabilistic and statistical model-checking on systems controlled by strategies. This will open the possibility of model checking real-time rewrite theories following the real-time strategies proposed here.

The paper [6] uses rewrite rules and “strategies” to analyze timed automata reachability using the rewriting framework ELAN [9]. In this setting, the authors define data types and rewrite rules for manipulating “zones” of the timed automaton, and then define rewrite strategies for various approaches to analyze these symbolic state spaces of the automaton. In other words: they use rewrite strategies not to restrict the possible behaviors of the timed automaton, but to define various analysis methods on the graph of zones of all behaviors of the automaton. In contrast, we define strategies to define and explore certain subsets of behaviors of timed systems.

## 7 Concluding Remarks

In this paper we propose what we hope is a useful yet reasonably intuitive language for defining execution strategies for real-time systems in Maude, allowing to perform most of the analysis methods supported by Real-Time Maude,

with user-defined strategies. Our language allows the user to define her discrete strategies and her time sampling strategies separately. We identify a number of interesting execution strategies for real-time systems, including a “mixed” time sampling strategy (not supported by Real-Time Maude) that should be ideal for explicit-state analysis of a large class of distributed real-time systems, such as our round trip time protocol.

Our strategies are given a semantics in Maude and are therefore implemented in Maude. A preliminary performance comparison between standard Maude search and our strategy-implemented reachability analyses on the CASH scheduling algorithm benchmark indicates that the latter are fairly competitive.

The benefits of this work are: (i) allowing the user to quickly and easily analyze her real-time system under a wide range of different scenarios without having to perform the error-prone task of modifying her model; (ii) providing much “better” (and flexible) time sampling strategies for time-sampling-based explicit-state analysis than provided by Real-Time Maude; (iii) providing a convenient framework in which we can quickly experiment with different strategies and analyses, before optimizing and hard-coding the most promising into the Real-Time Maude tool; (iv) allowing us to analyze real-time rewrite theories directly in Maude, instead of in Real-Time Maude, which is currently being re-developed; and (v) providing a backend for analyzing with user-defined strategies also time automata and time Petri nets, as well as for other modeling languages and formalisms for which Real-Time Maude provides a formal analysis backend.

Much work remains. For example, we should also provide untimed and timed temporal logic model checking combined with real-time strategies. For that, the `umaudemc` tool [36], and its support for model checking LTL and CTL formulas on strategy-controlled rewrite theories, will be fundamental. We also plan to combine *symbolic* analysis of real-time rewrite theories with user-defined strategies, as preliminary we have done in [2] and [4].

*Acknowledgments.* The authors acknowledge support from the NATO Science for Peace and Security Programme through grant number G6133 (project SymSafe) and the PHC project Aurora AESIR.

## References

1. AlTurki, M., Dhurjati, D., Yu, D., Chander, A., Inamura, H.: Formal specification and analysis of timing properties in software systems. In: *Fundamental Approaches to Software Engineering (FASE 2009)*. LNCS, vol. 5503, pp. 262–277. Springer (2009)
2. Arias, J., Bae, K., Olarte, C., Ölveczky, P.C., Petrucci, L., Rømming, F.: Rewriting logic semantics and symbolic analysis for parametric timed automata. In: *Proceedings of the 8th ACM SIGPLAN International Workshop on Formal Techniques for Safety-Critical Systems (FTSCS 2022)*. pp. 3–15. ACM (2022)
3. Arias, J., Bae, K., Olarte, C., Ölveczky, P.C., Petrucci, L., Rømming, F.: Symbolic analysis and parameter synthesis for networks of parametric timed automata with global variables using maude and smt solving. *Science of Computer Programming* (2023). <https://doi.org/10.1016/j.scico.2023.103074>, to appear

4. Arias, J., Bae, K., Olarte, C., Ölveczky, P.C., Petrucci, L., Rømming, F.: Symbolic analysis and parameter synthesis for time Petri nets using Maude and SMT solving. In: Application and Theory of Petri Nets and Concurrency (PETRI NETS 2023). LNCS, vol. 13929, pp. 369–392. Springer (2023). [https://doi.org/10.1007/978-3-031-33620-1\\_20](https://doi.org/10.1007/978-3-031-33620-1_20)
5. Bae, K., Ölveczky, P.C., Feng, T.H., Lee, E.A., Tripakis, S.: Verifying hierarchical Ptolemy II discrete-event models using Real-Time Maude. *Sci. Comput. Program.* **77**(12), 1235–1271 (2012)
6. Beffara, E., Bournez, O., Kacem, H., Kirchner, C.: Verification of timed automata using rewrite rules and strategies (2009)
7. Behrmann, G., David, A., Larsen, K.G.: A tutorial on Uppaal. In: Formal Methods for the Design of Real-Time Systems (SFM-RT 2004). LNCS, vol. 3185, pp. 200–236. Springer (2004)
8. Bobba, R., Grov, J., Gupta, I., Liu, S., Meseguer, J., Ölveczky, P.C., Skeirik, S.: Survivability: Design, formal modeling, and validation of cloud storage systems using Maude. In: Assured Cloud Computing, chap. 2, pp. 10–48. John Wiley & Sons (2018)
9. Borovanský, P., Kirchner, C., Kirchner, H., Ringeissen, C.: Rewriting with strategies in ELAN: A functional semantics. *Int. J. Found. Comput. Sci.* **12**(1), 69–95 (2001). <https://doi.org/10.1142/S0129054101000412>
10. Bravenboer, M., Kalleberg, K.T., Vermaas, R., Visser, E.: Stratego/XT 0.17. A language and toolset for program transformation. *Sci. Comput. Program.* **72**(1-2), 52–70 (2008). <https://doi.org/10.1016/J.SCICO.2007.11.003>
11. Caccamo, M., Buttazzo, G.C., Sha, L.: Capacity sharing for overrun control. In: Proceedings of the 21st IEEE Real-Time Systems Symposium (RTSS 2000). pp. 295–304. IEEE Computer Society (2000)
12. Clavel, M., Durán, F., Eker, S., Escobar, S., Lincoln, P., Martí-Oliet, N., Meseguer, J., Rubio, R., Talcott, C.: Maude Manual (Version 3.2.1). SRI International (2022), available at <http://maude.cs.illinois.edu>
13. Clavel, M., Durán, F., Eker, S., Lincoln, P., Martí-Oliet, N., Meseguer, J., Talcott, C.L.: All About Maude – A High-Performance Logical Framework, LNCS, vol. 4350. Springer (2007)
14. David, A., Jensen, P.G., Larsen, K.G., Mikucionis, M., Taankvist, J.H.: Uppaal Stratego. In: TACAS 2015. LNCS, vol. 9035. Springer (2015)
15. Eker, S., Martí-Oliet, N., Meseguer, J., Rubio, R., Verdejo, A.: The Maude strategy language. *J. Log. Algebraic Methods Program.* **134**, 100887 (2023). <https://doi.org/10.1016/J.JLAMP.2023.100887>
16. Grov, J., Ölveczky, P.C.: Formal modeling and analysis of Google’s Megastore in Real-Time Maude. In: Specification, Algebra, and Software – Essays Dedicated to Kokichi Futatsugi. LNCS, vol. 8373, pp. 494–519. Springer (2014)
17. Hidalgo-Herrero, M., Verdejo, A., Ortega-Mallén, Y.: Using Maude and its strategies for defining a framework for analyzing Eden semantics. In: Antoy, S. (ed.) WRS@FLoC 2006. ENTCS, vol. 174, pp. 119–137. Elsevier (2006). <https://doi.org/10.1016/J.ENTCS.2007.02.051>
18. Lepri, D., Ábrahám, E., Ölveczky, P.C.: Sound and complete timed CTL model checking of timed Kripke structures and real-time rewrite theories. *Sci. Comput. Program.* **99**, 128–192 (2015)
19. Liu, S., Meseguer, J., Ölveczky, P.C., Zhang, M., Basin, D.A.: Bridging the semantic gap between qualitative and quantitative models of distributed systems. *Proc. ACM Program. Lang.* **6**(OOPSLA2), 315–344 (2022)

20. Liu, S., Ölveczky, P.C., Meseguer, J.: Modeling and analyzing mobile ad hoc networks in Real-Time Maude. *J. Log. Algebraic Methods Program.* **85**(1), 34–66 (2016)
21. Marin, M., Kutsia, T.: Foundations of the rule-based system rLog. *J. Appl. Non Class. Logics* **16**(1-2), 151–168 (2006). <https://doi.org/10.3166/JANCL.16.151-168>
22. Meseguer, J.: Conditional rewriting logic as a unified model of concurrency. *Theor. Comput. Sci.* **96**(1), 73–155 (1992)
23. Meseguer, J.: Membership algebra as a logical framework for equational specification. In: *Recent Trends in Algebraic Development Techniques (WADT'97)*. LNCS, vol. 1376, pp. 18–61. Springer (1997)
24. Olarte, C., Ölveczky, P.C.: `rt-strategies` (2024), <https://depot.lipn.univ-paris13.fr/real-time-maude/rt-strategies.git>
25. Ölveczky, P.C.: Semantics, simulation, and formal analysis of modeling languages for embedded systems in Real-Time Maude. In: *Formal Modeling: Actors, Open Systems, Biological Systems – Essays Dedicated to Carolyn Talcott on the Occasion of Her 70th Birthday*, LNCS, vol. 7000, pp. 368–402. Springer (2011)
26. Ölveczky, P.C.: Real-Time Maude and its applications. In: *Rewriting Logic and Its Applications (WRLA 2014)*. LNCS, vol. 8663, pp. 42–79. Springer (2014)
27. Ölveczky, P.C., Boronat, A., Meseguer, J.: Formal semantics and analysis of behavioral AADL models in Real-Time Maude. In: *Formal Techniques for Distributed Systems, Joint 12th IFIP WG 6.1 International Conference, FMOODS 2010 and 30th IFIP WG 6.1 FORTE 2010*. LNCS, vol. 6117, pp. 47–62. Springer (2010)
28. Ölveczky, P.C., Caccamo, M.: Formal simulation and analysis of the CASH scheduling algorithm in Real-Time Maude. In: *Fundamental Approaches to Software Engineering (FASE 2006)*. LNCS, vol. 3922, pp. 357–372. Springer (2006)
29. Ölveczky, P.C., Meseguer, J.: Specification of real-time and hybrid systems in rewriting logic. *Theor. Comput. Sci.* **285**(2), 359–405 (2002)
30. Ölveczky, P.C., Meseguer, J.: Abstraction and completeness for Real-Time Maude. In: *6th International Workshop on Rewriting Logic and its Applications (WRLA 2006)*. *Electronic Notes in Theoretical Computer Science*, vol. 174, pp. 5–27. Elsevier (2006)
31. Ölveczky, P.C., Meseguer, J.: Semantics and pragmatics of Real-Time Maude. *High. Order Symb. Comput.* **20**(1-2), 161–196 (2007)
32. Ölveczky, P.C., Meseguer, J.: The Real-Time Maude tool. In: *Tools and Algorithms for the Construction and Analysis of Systems (TACAS 2008)*. LNCS, vol. 4963, pp. 332–336. Springer (2008)
33. Ölveczky, P.C., Meseguer, J., Talcott, C.L.: Specification and analysis of the AER/NCA active network protocol suite in Real-Time Maude. *Formal Methods Syst. Des.* **29**(3), 253–293 (2006)
34. Ölveczky, P.C., Thorvaldsen, S.: Formal modeling, performance estimation, and model checking of wireless sensor network algorithms in Real-Time Maude. *Theor. Comput. Sci.* **410**(2-3), 254–280 (2009)
35. Rosa-Velardo, F., Segura, C., Verdejo, A.: Typed mobile Ambients in Maude. In: Cirstea, H., Martí-Oliet, N. (eds.) *RULE@RDP 2005*. *ENTCS*, vol. 147, pp. 135–161. Elsevier (2005). <https://doi.org/10.1016/J.ENTCS.2005.06.041>
36. Rubio, R., Martí-Oliet, N., Pita, I., Verdejo, A.: Strategies, model checking and branching-time properties in maude. *J. Log. Algebraic Methods Program.* **123**, 100700 (2021). <https://doi.org/10.1016/J.JLAMP.2021.100700>
37. Rubio, R., Martí-Oliet, N., Pita, I., Verdejo, A.: Model checking strategy-controlled systems in rewriting logic. *Autom. Softw. Eng.* **29**(1), 7 (2022). <https://doi.org/10.1007/S10515-021-00307-9>

38. Rubio, R., Martí-Oliet, N., Pita, I., Verdejo, A.: Simulating and model checking membrane systems using strategies in Maude. *J. Log. Algebraic Methods Program.* **124**, 100727 (2022). <https://doi.org/10.1016/J.JLAMP.2021.100727>
39. Sabahi-Kaviani, Z., Khosravi, R., Ölveczky, P.C., Khamespanah, E., Sirjani, M.: Formal semantics and efficient analysis of Timed Rebeca in Real-Time Maude. *Sci. Comput. Program.* **113**, 85–118 (2015)
40. Santos-García, G., Palomino, M., Verdejo, A.: Rewriting logic using strategies for neural networks: An implementation in Maude. In: Corchado, J.M., Rodríguez, S., Llinas, J., Molina, J.M. (eds.) *DCAI 2008. Advances in Soft Computing*, vol. 50, pp. 424–433. Springer (2008). [https://doi.org/10.1007/978-3-540-85863-8\\_50](https://doi.org/10.1007/978-3-540-85863-8_50)